

Source:
ERICN Dartmouth Coll., Hanover, N.H. Kiewit Computaticn
Center.
AGENCY National Science Foundation, Washington, D.C.
12 May 70
NSF-GW-2246
52p.; For related documents, see SE 028 535-538; Not
available in hard copy due to copyright restrictions;
Computer print-out will not reproduce well

PRICE MF01 Plus Postage. PC Not Available from EDRS.
DESCRIPTORS *Business Education; Business Skills; *Computer
Oriented Programs; *Computer Programs; Computers;
Curriculum Guides; *Instruction; *Mathematics
Education; *Programing; Secondary Education

This teacher's guide is for a course designed to
teach business students the fundamentals of the BASIC language and
programming using a series of business-oriented programs.
The manual contains an introduction, flow charts, and computer
programs. The six lesson topics are print-out and format control,
file management, withholding tax schedule, payroll, three methods of
inventory control, and simple inventory control. (MP)

productions supplied by EDRS are the best that can be made *
from the original document. *

U.S. DEPARTMENT OF HEALTH,
EDUCATION & WELFARE
NATIONAL INSTITUTE OF
EDUCATION

THIS DOCUMENT HAS BEEN REPRODUCED EXACTLY AS RECEIVED FROM THE PERSON OR ORGANIZATION ORIGINATING IT. POINTS OF VIEW OR OPINIONS STATED DO NOT NECESSARILY REPRESENT OFFICIAL NATIONAL INSTITUTE OF EDUCATION POSITION OR POLICY.

"PERMISSION TO REPRODUCE THIS MATERIAL IN MICROFICHE ONLY HAS BEEN GRANTED BY

Mary L. Charles
NSF

TO THE EDUCATIONAL RESOURCES INFORMATION CENTER (ERIC)."

ED174464

SF 028 536

REC'D SCIS JUL 6 1970

A COMPUTER COURSE FOR BUSINESS STUDENTS

by

*Ann Waterhouse
South Portland High School*

TEACHER'S GUIDE

*SECONDARY SCHOOL PUBLICATION
KIEWIT COMPUTATION CENTER
DARTMOUTH COLLEGE
HANOVER, NEW HAMPSHIRE 03755*

*(Supported in part by National
Science Foundation Grant No.
GW-2246)*

12 May 1970

Copyright 1970 by Trustees of Dartmouth College

TABLE OF CONTENTS

INTRODUCTION	1
I PRINT-OUT AND FORMAT CONTROL	2
II COUNT-AVERAGE	11
III WITHHOLDING TAX SCHEDULE	13
IV PAYROLL	23
V THREE METHODS OF DEPRECIATION	30
VI SIMPLE INVENTORY CONTROL	39

INTRODUCTION

This course was first prepared for a group of seniors enrolled in Secretarial Officer Practice. Before the course began, six of the quicker students had mastered the unit record equipment through the accounting machine. These girls became our experimental class. The course was such a success that it was repeated in the spring for 8 more commercial students. These students were, in general, less able, but they completed the same material at a slightly slower pace.

The course lasts eight to ten weeks.

The objective of the course is to teach the students the fundamentals of the Basic language and computer programming using a series of business-oriented programs.

This entire unit, including on-line computer use, was completed within the regularly scheduled class period. The length of this unit could be shortened considerably by requiring outside assignments. This was not done for several reasons, not the least of which was the scarcity of computer time in the afternoon.

I. PRINT-OUT AND FORMAT CONTROL

(3 days)

Purpose: To introduce the six fundamental BASIC commands - READ, DATA, PRINT, END, IF-THEN, and GO TO and to illustrate the various print-out options including the TAB function.

The entire presentation of print-out was done using prepared acetates similar to the material in Figure 1.

- 1st day: Print-out through the use of the comma and semicolon.
- 2nd day: All students spent the time at the teletype writing various print-out programs
- 3rd day: Lecture on how to eliminate the OUT OF DATA statement, two-column print-out, use of the TAB function, and how to skip spaces in a program.

Figure 1

```
100 READ X
200 PRINT X
900 DATA 12,15,17,24,29,33,14,89,71,56,61,40,19,92,55
999END
```

LIST

```
DATA          15:27          09/08/67
```

```
100 READ X
200 PRINT X
900 DATA 12,15,17,24,29,33,14,89,71,56,61,40,19,92,55
999 END
```

READY

RUN

```
DATA          15:2          09/08/67
```

12

TIME: .04 SECS.

300 GO TO 100.

LIST

DATA 15:28 09/08/67

100 READ X
200 PRINT X
300 GO TO 100
900 DATA 12,15,17,24,29,33,14,89,71,56,61,40,19,92,55
999 END

READY

RUN

DATA 15:28 09/08/67

12
15
17
24
29
33
14
89
71
56
61
40
19
92
55
OUT OF DATA IN 100

TIME: .05 SECS.

200 PRINT X,

RUN

DATA 15:30 09/08/67

12	15	17	24	29
33	14	89	71	56
61	40	19	92	55

OUT OF DATA IN 100

TIME: .07 SECS.

200 PRINT X;

RUN

DATA 15:30 09/08/67

12	15	17	24	29	33	14	89	71	56	61	40	19	92	55
----	----	----	----	----	----	----	----	----	----	----	----	----	----	----

OUT OF DATA IN 100

TIME: .07 SECS.

```
950 DATA 0
150 IF X = 0 THEN 999
```

LIST

```
DATA          15:32          09/08/67
```

```
100READ X
150 IF X = 0 THEN 999
200 PRINT X;
300 GO TO 100
900 DATA 12,15,17,24,29,33,14,89,71,56,61,40,19,92,55
950 DATA 0
999END
```

READY

RUN

```
DATA          15:32          09/08/67
```

```
12 15 17 24 29 33 14 89 71 56 61 40 19 92 55
```

```
TIME:      .05 SECS.
```

100 READ X,Y

LIST

DATA 15:35 09/08/67

100 READ X,Y
150 IF X = 0 THEN 999
200 PRINT X;
300 GO TO 100
900 DATA 12,15,17,24,29,33,14,89,71,56,61,40,19,92,55
950 DATA 0
999 END

READY.

SAVE
READY

RUN

DATA 15:36 09/08/67

12 17 29 14 71 61 19 55
OUT OF DATA IN 100

TIME: .05 SECS.

```
150
250 IF Y = 0 THEN 999
```

```
LIST
```

```
DATA          15:40          09/08/67
```

```
100 READ X,Y
200 PRINT X;
250 IF Y = 0 THEN 999
300 GO TO 100
900 DATA 12,15,17,24,29,33,14,89,71,56,61,40,19,92,55
950 DATA 0
999 END
```

```
READY
```

```
RUN
```

```
DATA          15:41          09/08/67
```

```
12 17 29 14 71 61 19 55
```

```
TIME:      .05 SECS.
```

```
SAVE
FILE ALREADY SAVED !!
REPLACE? ANS: YES OR NO--YES
READY
```

200 PRINT X,Y

RUN

DATA 15:43 09/08/67

12	15
17	24
29	33
14	89
71	56
61	40
19	92
55	0

TIME: .08 SECS.

200 PRINT X;Y

RUN

DATA 15:43 09/08/67

12	15
17	24
29	33
14	89
71	56
61	40
19	92
55	0

TIME: .07 SECS.

```
200 PRINT X; TAB(30); Y
```

```
RUN
```

```
DATA          15:46          09/08/67
```

12	15
17	24
29	33
14	89
71	56
61	40
19	92
55	0

```
TIME:          .09 SECS.
```

```
210 PRINT
```

```
LIST
```

```
DATA          15:49          09/08/67
```

```
100 READ X,Y
200 PRINT X; TAB(30); Y
210 PRINT
250 IF Y = 0 THEN 999
300 GO TO 100
900 DATA 12,15,17,24,29,33,14,89,71,56,61,40,19,92,55
950 DATA 0
999 END
```

```
READY
```

II. COUNT - AVERAGE ROUTINE



(3 days)

Purpose: To write a simple, yet useful, program which would introduce the LET statement and illustrate how to print labels using quotation marks.

1st day: Construct the flow chart of the counting routine and then the summing routine beside each other as shown in Figure 2. The students can easily combine the two into a single flow chart for averaging a set of data. The students wrote the BASIC program from the flow chart in class.

2nd day: Each student typed in his average program and ran it with 3 sets of data. It is essential here that each student use block of data of varying lengths. There are hazards here, particularly if the next set of numbers fills less lines of data than the previous block did.

3rd day: Continuation of the second day.



III. WITHHOLDING TAX TABLE

(about 6 days)

Purpose: To study simple and nested loops and the use of FOR - NEXT, INPUT, the INT function and the DEF instruction.

Acetates were prepared showing income tax withholding by the percentage method, the general directions and a sample page from the table for a weekly pay period. These acetates are not made from the latest Employers Tax Guide. An older version is more suitable for teaching looping. The attached sequence of programs (Figure 3) shows the progress made by the class, first printing out the amount to be withheld for 0 through 6 exemptions for a given wage of \$40, then trying to duplicate the tax table entries and, finally, a portion of the actual table itself. The students developed the initial flow chart, wrote, and ran the program immediately. The results were viewed immediately using the overhead projector. They discussed the things which needed to be changed, returned to the teletype, and tried again. The entire week was spent in this manner. Note that the problems encountered led naturally to the following development:

1. Need for a variable TAB function to space the columns.
2. Introduction of the FOR - NEXT commands
3. Elimination of the negative values of withholding tax because \$13 times the number of exemptions exceeded the gross pay.
4. Introduction of the INPUT command to facilitate varying the gross pay during the RUN of the program.
5. Necessity to round off to the nearest hundredth.
6. Double loop to print out a block of the actual table.

An interesting sidelight to this whole project was the surprise of the students when they discovered that it was practically impossible to duplicate the table printed by the government.

Figure 3

LIST

WITHOLD 15:38 03/01/68

```

100 READ G
200 DATA 40
300 LET N=0
400 LET T=(G-13*N)*.14
450 PRINT G, T;TAB (24);
500 LET N = N+1
600 IF N > 6 THEN 800
700 GO TO 400
800 END

```

READY

RUN

WITHOLD 15:39 03/01/68

40	5.6	40	3.78	40	1.96	40	0.14	40
-1.68		40	-3.5	40	-5.32			

TIME: .05 SECS.

40 5.6 3.78 1.96 0.14 -1.68 -3.5 -5.32
TIME: .07 SECS.

450 PRINT T;TAB(22+N*7);

RUN

WITHOLD 15:41 03/01/68

40 5.6 3.78 1.96 0.14 -1.68 -3.5 -5.32
TIME: .07 SECS.

19

LIST

WITHOLD 17:08 01/05/68

```
100 READ G
200 DATA 40
250 PRINT G,
300 FOR N = 0 TO 6
400 LET T = (G-13*N)*.14
450 PRINT T;TAB(22+N*7);
500 NEXT N
800 END
```

READY

RUN

WITHOLD 17:09 01/05/68

40 5.6 3.78 1.96 0.14 -1.68 -3.5 -5.32

TIME: .04 SECS.

LIST

WITHOLD 17:37 01/05/68

```
100 READ G
200 DATA 40
250 PRINT G,
300 FOR N = 0 to 6
400 LET T = (G-13*N)*.14
425 IF T > 0 then 450
430 PRINT 0;TAB(22+N*7);
440 GO TO 500
450 PRINT T;TAB(22+N*7);
500 NEXT N
800 END
```

READY

RUN

WITHOLD 17:37 01/05/68

40 5.6 3.78 1.96 0.14 0 0 0

TIME: .04 SECS.

LIST

WITHOLD 17:35 01/05/68

```
90 PRINT "WHAT IS THE GROSS PAY";
100 INPUT G
110 PRINT
250 PRINT G,
300 FOR N = 0 TO 6
400 LET T = (G-13*N)*.14
425 IF T > 0 THEN 450
430 PRINT 0;TAB(22+N*7);
440 GO TO 500
450 PRINT T;TAB(22+N*7);
500 NEXT N
800 END
```

READY

RUN

WITHOLD 17:33 01/05/68

WHAT IS THE GROSS PAY? 40
 40 5.6 3.78 1.96 0.14 0 0 0

TIME: .09 SECS.

110 PRINT
 RUN

WITHOLD 17:34 01/05/68

WHAT IS THE GROSS PAY? 40.5
 40.5 5.67 3.85 2.03 0.21 0 0 0

TIME: .08 SECS.

RUN

WITHOLD 17:34 01/05/68

WHAT IS THE GROSS PAY? 40.7
 40.7 5.698 3.878 2.058 0.238 0 0 0

TIME: .05 SECS.

RUN

WITHOLD 17:35 01/05/68

WHAT IS THE GROSS PAY? 40.8
 40.8 5.712 3.892 2.072 0.252 0 0 0

TIME: .05 SECS.

LIST

WITHOLD 17:30 01/05/68

```

90 FOR G = 48.8 TO 60.8
250 PRINT G,
300 FOR N = 0 TO 6
400 LET T = (G-13*N)*.14
425 IF T > 0 THEN 450
430 PRINT 0;TAB(22+N*7);
440 GO TO 500
450 PRINT T;TAB(22+N*7);
500 NEXT N
550 PRINT
600 NEXT G
800 END

```

READY

RUN

WITHOLD 17:31 01/05/68

48.8	6.832	5.012	3.192	1.372	0	0	0
49.8	6.972	5.152	3.332	1.512	0	0	0
50.8	7.112	5.292	3.472	1.652	0	0	0
51.8	7.252	5.432	3.612	1.792	0	0	0
52.8	7.392	5.572	3.752	1.932	0.112	0	0
53.8	7.532	5.712	3.892	2.072	0.252	0	0
54.8	7.672	5.852	4.032	2.212	0.392	0	0
55.8	7.812	5.992	4.172	2.352	0.532	0	0
56.8	7.952	6.132	4.312	2.492	0.672	0	0
57.8	8.092	6.272	4.452	2.632	0.812	0	0
58.8	8.232	6.412	4.592	2.772	0.952	0	0
59.8	8.372	6.552	4.732	2.912	1.092	0	0
60.8	8.512	6.692	4.872	3.052	1.232	0	0

TIME: .13 SECS.

LIST

WITHOLD 17:45 01/05/68

```
80 DEF FNR(X) = INT(X*10+.5)/10
90 FOR G = 40.8 TO 50.8
250 PRINT G,
300 FOR N = 0 TO 6
400 LET T = (G-13*N)*.14
425 IF T > 0 THEN 450
430 PRINT 0;TAB(22+N*7);
440 GO TO 500
450 PRINT FNR(T);TAB(22+N*7);
500 NEXT N
550 PRINT
600 NEXT G
800 END
```

READY

WITHOLD 17:42 01/05/68

48.8	6.8	5	3.2	1.4	0	0	0
49.8	7	5.2	3.3	1.5	0	0	0
50.8	7.1	5.3	3.5	1.7	0	0	0
51.8	7.3	5.4	3.6	1.8	0	0	0
52.8	7.4	5.6	3.8	1.9	0.1	0	0
53.8	7.5	5.7	3.9	2.1	0.3	0	0
54.8	7.7	5.9	4	2.2	0.4	0	0
55.8	7.8	6	4.2	2.4	0.5	0	0
56.8	8	6.1	4.3	2.5	0.7	0	0
57.8	8.1	6.3	4.5	2.6	0.8	0	0
58.8	8.2	6.4	4.6	2.8	1	0	0
59.8	8.4	6.6	4.7	2.9	1.1	0	0
60.8	8.5	6.7	4.9	3.1	1.2	0	0

TIME: .17 SECS.

90 FOR G = 40.8 TO 50.8

RUN

WITHOLD 17:44 01/05/68

40.8	5.7	3.9	2.1	0.3	0	0	0
41.8	5.9	4	2.2	0.4	0	0	0
42.8	6	4.2	2.4	0.5	0	0	0
43.8	6.1	4.3	2.5	0.7	0	0	0
44.8	6.3	4.5	2.6	0.8	0	0	0
45.8	6.4	4.6	2.8	1	0	0	0
46.8	6.6	4.7	2.9	1.1	0	0	0
47.8	6.7	4.9	3.1	1.2	0	0	0
48.8	6.8	5	3.2	1.4	0	0	0
49.8	7	5.2	3.3	1.5	0	0	0
50.8	7.1	5.3	3.5	1.7	0	0	0

TIME: .15 SECS.

IV. PAYROLL

(2 to 3 weeks)

Purpose: To print out the weekly payroll register of a small local company employing 10 people.

1. Preliminaries

The students were given the master list of employees, Figure 4. This contains name, payroll number, marital status, number of claimed exemptions, and hourly rate (or salary). Subsequent discussion of how the various items of data should be designated led to the introduction of subscripted variables, use of the \$ for alphanumeric data, such as name and payroll number, and study of the MAT INPUT command which would allow changing the hours worked each week as the program runs.

2. Flow Chart

The flow chart took one week and was done in two parts because of the complexity of the withholding sequence. Figure 5 shows the first flow chart minus the detailed withholding routine. The following is the key for interpreting the various letters used.

G = Gross pay	A\$(I) = Ith person's name
S = Social Security	B\$(I) = Ith person's payroll number
W = Withholding tax	N(I) = Number of claimed exemptions
W1 = The amount of gross pay which withholding tax is computed	M(I) = Marital status
P = Net pay	V = Hours of overtime
	H(I) = Hours worked, Ith person (input a 0 if salaried)

The decision box containing $S(I) > 5$ checks to see whether the person is salaried or not since the letter S can represent either salary or hourly rate in this program.

To save time the flow chart of the withholding routine was prepared in advance. Two overhead projectors were procured and the flow chart was displayed alongside a current Weekly Table for the Percentage Method of Withholding, Figure 6. The flow chart was compared, step-by-step, with the table. Once the student understood the patterns involved, they wrote the BASIC program very easily and inserted it into the main payroll routine.

3. Work Period


An additional 2 weeks were necessary for all students to debug their programs and insert the appropriate column heading and spacings, Figure 7.


Figure 4

LIST 09:19 01/13/68

9000	DATA	V. CLOUTIER, "999-23", 1,1,1.50
9010	DATA	K. EATON, "999-11", 2,2,110.00
9020	DATA	K. GARTHWAITE, "999-31", 1,0,1.00
9030	DATA	D. MART, "999-15", 1,0,1.75
9040	DATA	R. HORNING, "999-26", 1,1,75.00
9050	DATA	A. MILLER, "999-24", 1,1,1.50
9060	DATA	J. MILLER, "999-29", 1,1,1.40
9070	DATA	S. OWEN, "999-28", 1,1,1.60
9080	DATA	C. TWOMBLY, "999-19", 1,1,1.50
9090	DATA	T. WEST, "999-13", 2,0,110.00

Figure 5





PAYRØLL1

```

90 DEF FNF(X) = INT(X*100+.5)/100
100 MAT INPUT H(10)
101 PRINT
102 PRINT
103 PRINT TAB(13);"PAYRØLL";
104 PRINT TAB(36);"GRØSS";TAB(46);"WITH";TAB(63);" NET"
105 PRINT " NAME";TAB(13);"NUMBER";TAB(21);" SALARY";
106 PRINT TAB(36);" PAY";TAB(46);" TAX";TAB(56);"FICA";
107 PRINT TAB(63);" PAY"
110 PRINT
120 PRINT
200 FØR I = 1 TØ 10
300 READ A$(I),B$(I),M(I),N(I),S(I)
400 IF S(I) <5 THEN 700
500 LET G = S(I)
600 GØ TØ 1200
700 IF H(I)>48 THEN 1000
800 LET G = H(I)*S(I)
900 GØTØ 1200
1000 LET V = H(I) - 48
1100 LET G = 48*S(I) +V*S(I)*1.5
1200 LET W1 = G-13.50*N
1300 IF M(I) >1 THEN 1600
1310 IF W1>4 THEN 1340
1320 LET W = 0
1330 GØTØ 1800
1340 IF W1>13 THEN 1370
1350 LET W = (W1-4)*.14
1360 GØTØ 1800
1370 IF W1>23 THEN 1400
1380 LET W = 1.26+(W1-13)*.15
1390 GØTØ 1800
1400 IF W1>85 THEN 1430
1410 LET W = 2.76+(W1-23)*.17
1420 GØTØ 1800
1430 IF W1>169 THEN 1460
1440 LET W = 13.30+(W1-85)*.20
1450 GØTØ 1800
1460 IF W1>212 THEN 1490
1470 LET W = 30.10+(W1-169)*.25
1480 GØTØ 1800
1490 LET W = 40.84+(W1-212)*.30
1500 GØTØ 1800
1600 IF W1>4 THEN 1630
1610 LET W = 0
1620 GØTØ 1800
1630 IF W1>23 THEN 1660
1640 LET W = (W1-4)*.14
1650 GØTØ 1800
1660 IF W1 > 85 THEN 1690

```

PAYROLL1 (CONTINUED)

```

1670 LET W = 2.66+(W1-23)*.15
1680 GOTO 1800
1690 IF W1 > 169 THEN 1720
1700 LET W = 11.96+(W1-85)*.17
1710 GOTO 1800
1720 IF W1 > 340 THEN 1750
1730 LET W = 26.24 + (W1-169)*.20
1740 GOTO 1800
1750 IF W1 > 423 THEN 1780
1760 LET W = 60.44 + (W1-340)*.25
1770 GOTO 1800
1780 LET W = 31.19 + (W1-423)*.30
1800 LET S = G * .044
1900 LET P = G - W - S
2000 IF S(I) > 5 THEN 2400
2100 PRINT A$(I);TAB(13);B$(I);TAB(21);H(I)"HR @"S(I);
2200 PRINT TAB(35);FNF(G);TAB(45);FNF(W);TAB(55);FNF(S);TAB(63);FNF(P)
2250 PRINT
2300 GOTO 2600
2400 PRINT A$(I);TAB(13);B$(I);TAB(20);" ";S(I);
2500 GOTO 2200
2600 NEXT I
9000 DATA V.CLOUTIER,"999-23",1,1,1.50
9010 DATA K.EATON,"999-11",2,2,110.00
9020 DATA K.GARTHWAITE,"999-31",1,0,1.00
9030 DATA D.HART,"999-15",1,0,1.75
9040 DATA R.HORNING,"999-26",1,1,75.00
9050 DATA A.MILLER,"999-25",1,1,1.50
9060 DATA J.MILLER,"999-29",1,1,1.40
9070 DATA S.OWEN,"999-28",1,1,1.60
9080 DATA C.TWOMBLY,"999-19",1,1,1.50
9090 DATA T.WEST,"999-13",2,0,110.00
9999 END

```

RUN

PAYROLL1 12 MAY 70 13:27

? 30,0,4,50,0,23,23,39,39,0

NAME	PAYROLL NUMBER	SALARY	GROSS PAY	WITH TAX	FICA	NET PAY
V. CLOUTIER	999-23	30 HR @ 1.5	45	6.5	1.98	36.52
K. EATON	999-11	110	110	16.21	4.84	88.95
K. GARTHWAITE	999-31	4 HR @ 1	4	0	0.18	3.82
D. HART	999-15	50 HR @ 1.75	89.25	14.15	3.93	71.17
R. HORNING	999-26	75	75	11.6	3.3	60.1
A. MILLER	999-25	23 HR @ 1.5	34.5	4.71	1.52	28.27
J. MILLER	999-29	23 HR @ 1.4	32.2	4.32	1.42	26.46
S. OWEN	999-28	39 HR @ 1.6	62.4	9.46	2.75	50.2
C. TWOMBLY	999-19	39 HR @ 1.5	58.5	8.79	2.57	47.13
T. WEST	999-13	110	110	16.21	4.84	88.95

TIME: 0.465 SEC.
READY

V. DEPRECIATION USING THE THREE COMMON METHODS

(1-2 weeks)

Purpose: To utilize the programming techniques already learned such as looping, labeling, and rounding off.

Note: These students had studied the three methods of depreciation in Bookkeeping II; therefore, this section was essentially review.

1st day: Straight-line method. The class carried out a pencil calculation of a simple example, say \$1500 for 5 years. From this example they designed the flow chart.

2nd day: The students spent the period on the computer perfecting their programs, labeling, etc.


3rd day: Double-declining balance method. Again the class carried out a long-hand calculation of a simple example and then wrote the flow chart.

4th day: The students used the computer to perfect their programs.

5th day: Sum-of-the-years-digits method. After working through a numerical example, the first group of students attempted to write the program unassisted. They spent three days working on the computer struggling to duplicate the original example. They had considerable difficulty because the annual depreciation is figured by applying a changing fraction to the original cost of the asset. As a result, the second group was taught this method in the same fashion as the previous two types.

When each student has a complete set of depreciation programs, the following problems could be assigned. Some problems make use of salvage value, which is a useful refinement of the original programs, Figure 8. If the class is a good one, Part I will present a challenge. This is not an easy program. One good program is shown in Figure 9. It took a week of effort by the best programmer. (This part was not even suggested for the second class.)

DOUBLE-DECLINING METHOD



W = worth of asset
 Y = number of years
 D = depreciation
 B = book value

SUM-OF-THE-YEARS-DIGITS




Figure 8

LIS

STRA-LIN 12 MAY 70 13:31

80 REM THIS COMPUTES DEPRECIATION BY THE STRAIGHT-LINE METHOD.

90 DEF FNR(X) = INT(X*100+.5)/100

100 INPUT W,Y

110 PRINT

120 PRINT TAB(31);"BOOK"

130 PRINT "YEAR";TAB(16);"DEPR.";TAB(31);"VALUE"

140 PRINT

150 PRINT

160 LET D = W/Y

170 FOR I = 1 TO Y

180 LET B = W-D*I

190 PRINT I, FNR(D), FNR(B)

200 NEXT I

210 END

READY

RUN

STRA-LIN 12 MAY 70 13:32

? 10000,12

YEAR	DEPR.	BOOK VALUE
1	833.33	9166.67
2	833.33	8333.33
3	833.33	7500
4	833.33	6666.67
5	833.33	5833.33
6	833.33	5000
7	833.33	4166.67
8	833.33	3333.33
9	833.33	2500
10	833.33	1666.67
11	833.33	833.33
12	833.33	0

TIME: 0.155 SEC.

READY

LIS

DMU-DECL 12 MAY 70 13:34

```

80 REM THIS COMPUTES DEPRECIATION BY THE DOUBLE-DECLINING METHOD.
90 DEF FNR(X) = INT(X*100+.5)/100
100 INPUT W,Y
110 PRINT
120 PRINT TAB(31);"BOOK"
130 PRINT "YEAR";TAB(16);"DEPR.";TAB(31);"VALUE"
140 PRINT
150 PRINT
160 FOR I = 1 TO Y
170 LET D = 2/Y*W
180 LET B = W-D
190 PRINT I, FNR(D), FNR(B)
200 LET W = B
210 NEXT I
220 END
READY

```

RUN

DMU-DECL 12 MAY 70 13:35

? 10000,6

YEAR	DEPR.	BOOK VALUE
1	3333.33	6666.67
2	2222.22	4444.44
3	1481.48	2962.96
4	987.65	1975.31
5	658.44	1316.87
6	438.96	877.91

TIME: 0.117 SEC.

READY

LIS

SUM 12 MAY 70 13:36

```

80 REM THIS COMPUTES DEPRECIATION BY THE SUN-OF-THE-YEARS-DIGITS METHOD.
90 DEF FNR(X) = INT(X*100+.5)/100
100 INPUT W,Y
110 PRINT
120 PRINT TAB(31);"BOOK"
130 PRINT "YEAR";TAB(16);"DEPR.";TAB(31);"VALUE"
140 PRINT
150 PRINT
160 LET S = 0
170 FOR Z = 1 TO Y
180 LET S = S+Z
190 NEXT Z
200 LET B = W
210 FOR I = Y TO 1 STEP -1
220 LET D = I/S*W
230 LET B = B-D
240 PRINT Y-I+1, FNR(D), FNR(B)
250 NEXT I
260 END
READY

```

RUN

SUM 12 MAY 70 13:37

? 3600,5

YEAR	DEPR.	BOOK VALUE
1	1200	2400
2	960	1440
3	720	720
4	480	240
5	240	0

TIME: 0.121 SEC.
READY

I. Write a program which will print out a Comparison Chart of the three depreciation methods we have studied. Take into account the salvage value (amount for which an asset is sold when its useful life is ended. An asset with salvage value will have depreciation expense equal to its total cost less its salvage value.) For each method print only the depreciation and book value. Too many columns are necessary otherwise. Break your INPUT statement into three parts and have the computer request (by PRINT statements) the total costs of the asset, the salvage value and the estimated life in years.

Suggestions: Put the labels and headings on last. The main part of your program can be formed by attaching the working part of each of the three programs you have done in sequence. If you do this, you must use subscripts and you will need to change the letters in two of your programs. The print out will be the very last part of your program - separate loop. Or, you might observe that all three programs have an "I" loop from 1 to Y (the sum-of-the-years-digits-method can be written this way, too. In this case, the depreciation must be calculated using $(Y-I+1)$ instead of I). Thus, you should be able to calculate all three methods inside of one loop. Be sure to change the letters in two programs. The print out will be inside of the I loop and you will not need subscripts. Choose a method and stick with it.

II. 1. A building contractor bought a cement mixer for \$4,500. Its estimated life was 6 years with a junk value of \$300 at the end of that time. At the end of 4 years the machine was worn out and was sold for junk for \$225. How much difference was there between the book value and the cash value of the machine on the date of the sale.

(a) By straight-line method.

Ans.

(b) By sum-of-years-digits method.

Ans.

2. The Pyrotechnic Manufacturing Company bought a fire engine to use for the protection of their plant and paid \$21,000 for it. The engine was estimated to have a life of 20 years and a trade-in value at the end of that time for \$3,000. After 6 years, the company went out of business and sold the engine to a local fire company for \$10,000.

(a) If the machine was depreciated on the declining-balance plan, how much did the company lose on the sale? Ans.

(b) If the sum-of-the-years-digits plan had been used, would this loss have been smaller or greater, and how much? Ans.

3. The machinery purchased by the Kansota Corporation when it started operations cost \$69,600. It was decided to depreciate it by the declining-balance method over an estimated life of 16 years.

(a) What was the book value at the end of the 4th year? Ans.

(b) What was the total depreciation at the end of the 6th year? Ans.

4. A, B, and C have formed a partnership in a contracting business and have purchased grading equipment for \$36,000. In discussing how to depreciate it, the partners disagree. A says the equipment will be worn out in 12 years with no salvage value. B and C think it will last only 10 years, but that it will have resale value of 10% of cost. However, B favors the declining-balance method, while C wants to use the sum-of-the-years-digits plan. The partners wish to take the largest possible amount as a tax deduction for depreciation during the next five years. Show the three plans in parallel form and compare the totals to find which plan yields the greatest depreciation.

Figure 9

LIS

COMBINE 12 MAY 70 13:47

```

50 DIM B(40),E(40),Ø(40),P(40),K(40)
90 DEF FNR(X) = INT(X*100+.5)/100
100 INPUT W,Y,S
101 PRINT
102 PRINT
104 PRINT TAB(43);"SUM-ØF-THE-"
105 PRINT TAB(8);"STRAIGHT-LINE";
106 PRINT TAB(25);"DECLINING-BALANCE";
108 PRINT TAB(48);"YEAR-DIGITS"
109 PRINT
120 PRINT "XEAR";TAB(7);"DEPR.";TAB(17);"BØØK";
121 PRINT TAB(27);"DEPR.";TAB(37);"BØØK";
122 PRINT TAB(47);"DEPR.";TAB(57);"BØØK"
130 PRINT
140 LET W1 = W-S
150 LET D = W1/Y
160 FØR I = 1 TØ Y
170 LET B(I) = W-D*I
180 NEXT I
190 FØR I = 1 TØ Y
200 LET E(I) = 2/Y*W
210 LET Ø(I) = W-E(I)
220 LET W = Ø(I)
230 NEXT I
240 LET R = 0
350 FØR Z = 1 TØ Y
360 LET R = R+Z
370 NEXT Z
390 LET K(0) = W1 + S
400 FØR I = 1 TØ Y
410 LET P(I) = (Y-I+1)/R*W1
420 LET K(I) = K(I-1)-P(I)
430 NEXT I
440 FØR J = 1 TØ Y
450 PRINT J;TAB(6);FNR(D);TAB(16);FNR(B(J));TAB(26);FNR(E(J));
460 PRINT TAB(36);FNR(Ø(J));TAB(46);FNR(P(J));TAB(56);FNR(K(J))
470 NEXT J
900 END
READY

```

RUN

COMBINE 12 MAY 70 13:50

? 25000,20,3000

YEAR	STRAIGHT-LINE		DECLINING-BALANCE		SUM-OF-THE-YEAR-DIGITS	
	DEPR.	BOOK	DEPR.	BOOK	DEPR.	BOOK
1	1100	23900	2500	22500	2095.24	22904.8
2	1100	22800	2250	20250	1990.48	20914.3
3	1100	21700	2025	18225	1885.71	19028.6
4	1100	20600	1822.5	16402.5	1780.95	17247.6
5	1100	19500	1640.25	14762.2	1676.19	15571.4
6	1100	18400	1476.22	13286.	1571.43	14000
7	1100	17300	1328.6	11957.4	1466.67	12533.3
8	1100	16200	1195.74	10761.7	1361.9	11171.4
9	1100	15100	1076.17	9685.51	1257.14	9914.29
10	1100	14000	968.55	8716.96	1152.38	8761.9
11	1100	12900	871.7	7845.26	1047.62	7714.28
12	1100	11800	784.53	7060.74	942.86	6771.43
13	1100	10700	706.07	6354.66	838.1	5933.33
14	1100	9600	635.47	5719.2	733.33	5200
15	1100	8500	571.92	5147.28	628.57	4571.43
16	1100	7400	514.73	4632.55	523.81	4047.62
17	1100	6300	463.26	4169.3	419.05	3628.57
18	1100	5200	416.93	3752.37	314.29	3314.28
19	1100	4100	375.24	3377.13	209.52	3104.76
20	1100	3000	337.71	3039.42	104.76	3000

TIME: 0.442 SEC.
READY

VI. SIMPLE INVENTORY CONTROL

(1-2 weeks)

Purpose: To simulate the receipt and issuance of merchandise on a stockroom card and to keep stock record cards for a number of different items for a monthly period.

To introduce the use of double subscripts.

The program that resulted from this unit is designed to solve Jobs 45-48 in the text, Clerical Record Keeping, by Baron and Steinfeld (S W Publications, 1965). The user can enter the receipt and the issuance of various items of stock on a daily basis. The program is designed to notify the user if a particular item goes above the more an employer should have on hand or below the least he must have on hand to meet current demand. The user also has the option of printing out a complete stock record card showing the daily entries for the month or of listing a summary of all items in stock with the current balance.

The first group of students developed the flow chart in Figure 10 as a joint class effort over a period of three days. The following is the key for interpreting the various variables used in the chart.

$SS(I)$ = the stock number
 $US(I)$ = the unit (doz., each, etc.)
 $X(I)$ = maximum amount that should be on hand
 $M(I)$ = minimum amount that must be on hand
 $F(I)$ = balance carried forward from the previous month
 $D\$$ = the month
 J = the day of the month on which an entry is made
 Y = the year
 $A\$$ = the stock number of the record card to be changed
 Q = the quantity received (+) or issued (-)
 $D(I, J)$ = the date an entry is made for the Ith card
 $R(I, J)$ = the entry in the "received" column, Ith card, Jth day
 $S(I, J)$ = the entry in the "issued" column, Ith card, Jth day
 $B(I, J)$ = the balance on the Ith card, Jth day
 R = the total quantity received in a month
 S = the total quantity issued in a month
 $L(I)$ = the current balance for the Ith card. At the end of the month this value becomes the balance brought forward.

Unfortunately, the ranking period closed before the girls finished writing the actual program. This would have taken at least four additional days. The program which is attached shows what time did not permit. The second group of students did not have time to even begin this program.







Figure 10







STOCK

```

100 DIM S$(20),U$(20),B(20,32),D(20,32),R(20,32),S(20,32)
110 DIM L(20),X(20),M(20)
120 READ N,D$,Y
130 FOR I = 1 TO N
140 READ S$(I),U$(I),X(I),M(I),L(I)
150 LET F(I) = L(I)
160 NEXT I
170 PRINT "CHANGES IN INVENTORY";
180 INPUT W$
190 IF W$ = "NØ" THEN 440
200 INPUT J,A$,Q
210 FOR I = 1 TO N
220 IF S$(I) = A$ THEN 260
230 NEXT I
240 PRINT "THERE IS NØ CARD FOR STOCK NØ. "A$
250 GØTØ 170
260 LET D(I,J) = J
270 IF Q > 0 THEN 300
280 LET S(I,J) = -Q
290 GØTØ 310
300 LET R(I,J) = Q
310 LET B(I,J) = L(I)+Q
320 LET L(I) = B(I,J)
330 IF B(I,J) <=X(I) THEN 360
340 PRINT "BALANCE IS " B(I,J)-X(I)"ØVER MAXIMUM."
350 GØTØ 410
360 IF B(I,J) >= M(I) THEN 390
370 PRINT "BALANCE IS" M(I)-B(I,J)"BELOW MINIMUM."
380 GØTØ 410
390 IF B(I,J)-M(I) > 5 THEN 410
400 PRINT "TIME TØ REØRDER "A$
410 PRINT "ANY MØRE CHANGES";
420 INPUT Q$
430 IF Q$ = "YES" THEN 200
440 PRINT "DØ YØU WANT A PARTICULAR CARD";
450 INPUT Q$
460 IF Q$ = "NØ" THEN 740
470 PRINT "WHICH CARD";
480 INPUT A$
490 FOR I = 1 TO N
500 IF S$(I) = A$ THEN 540
510 NEXT I
520 PRINT "THERE IS NØ CARD FOR STOCK NP. "A$
530 GØTØ 440
540 PRINT
550 PRINT "DATE";TAB(11);"REC.";TAB(21);"ISS.";TAB(31);"BAL."
562 PRINT Y
570 PRINT D$;TAB(30);F(I)
580 LET R = S = 0
590 FOR K = 1 TO 31

```

STOCK (CONTINUED)

```

600 IF D(I,K) = 0 THEN 670
610 IF R(I,K) = 0 THEN 640
620 PRINT TAB(4);D(I,K);TAB(10);R(I,K);TAB(30);B(I,K)
630 GOTO 650
640 PRINT TAB(4);D(I,K);TAB(20);S(I,K);TAB(30);B(I,K)
650 LET R = R+R(I,K)
660 LET S = S+S(I,K)
670 NEXT K
680 PRINT
690 PRINT "TOTALS";TAB(10);R;TAB(20);S
700 PRINT
710 PRINT "ANOTHER CARD";
720 INPUT FS
730 IF FS = "YES" THEN 470
740 PRINT "DO YOU WANT A SUMMARY OF ALL CARDS";
750 INPUT FS
760 IF FS = "NO" THEN 960
770 PRINT
780 PRINT "STOCK #", " UNIT", " MAX.", " MIN.", " BAL."
790 PRINT
800 FOR I = 1 TO N
860 PRINT SS(I),US(I),X(I),M(I),L(I)
870 NEXT I
880 DATA 7,MAY,1968
890 DATA HB14,DOZEN,150,25,0
900 DATA VB4,DOZEN,300,75,93
910 DATA BAB14,DOZEN,850,150,144
920 DATA FR9,DOZEN,1800,250,528
930 DATA SH8,DOZEN,1200,300,274
940 DATA T310,DOZEN,400,75,106
950 DATA B36,DOZEN,350,60,64
960 END

```

RUN

STOCK 12 MAY 70 14:09

CHANGES IN INVENTORY? YES

? 2,VB4,150

ANY MORE CHANGES? YES

? 2,BAB14,625

ANY MORE CHANGES? YES

? 2,BB6,264

ANY MORE CHANGES? YES

? 5,SHS,1000

BALANCE IS 74 OVER MAXIMUM.

ANY MORE CHANGES? YES

? 5,VB4,-96

ANY MORE CHANGES? YES

? 5,FR9,1248

ANY MORE CHANGES? YES

? 5,TB10,-48

BALANCE IS 17 BELOW MINIMUM.

ANY MORE CHANGES? YES

? 9,SHS,-480

ANY MORE CHANGES? YES

? 9,YB10,296

THERE IS NO CARD FOR STOCK NO. YB10

CHANGES IN INVENTORY? YES

? 9,TB10,296

ANY MORE CHANGES? YES

? 9,BB6,-150

ANY MORE CHANGES? YES

? 14,BAB14,-274

ANY MORE CHANGES? 16,←←←YES

? 15,VB4,168←←←-78

BALANCE IS 6 BELOW MINIMUM.

ANY MORE CHANGES? YES

? 20,FR9,-720

ANY MORE CHANGES? YES

? 23,VB4,160

ANY MORE CHANGES? 24 ←←←YES

? 24,HB14,110

ANY MORE CHANGES? YES

? 26,HB14,-73

TIME TO REORDER HB14

ANY MORE CHANGES? NO

DO YOU WANT A PARTICULAR CARD? YES
WHICH CARD? VB4

DATE	REC.	ISS.	BAL.
1963			
MAY			93
2	150		243
5		95	147
16		73	69
23	168		237
TOTALS	318	174	

ANOTHER CARD? YES
WHICH CARD? SH8

DATE	REC.	ISS.	BAL.
1968			
MAY			274
5	1000		1274
9		486	788
TOTALS	1000	486	

ANOTHER CARD? NO
DO YOU WANT A SUMMARY OF ALL CARDS? YES

STOCK #	UNIT	MAX.	MIN.	BAL.
HB14	DOZEN	150	25	27
VB4	DOZEN	300	75	237
BAB14	DOZEN	850	150	495
FR9	DOZEN	1800	250	1056
SH8	DOZEN	1200	300	788
TB10	DOZEN	400	75	354
B36	DOZEN	350	50	172

TIME: 0.848 SEC.
READY